
1 Introduction

The floating-point architecturdor a processor consists of the different aspects thattaffe@ programs
operating on floating-point data are mapped onto the magcimaleding:

e How floating-point values are stored and accessed. Thipisaly via some form of registers.
e The instructions that operate on floating-point data.

e The conventions used for passing floating-point values @snagnts to functions, and for returning
them as results.

In this document, we will describe the floating-point arebitire for x86 processors known&SE

Since the introduction of the Pentium MMX in 1997, both Indéeld AMD have incorporated successive
generations ofmediainstructions to support graphics and image processingtir8avith the Pentium Il in
1999, these instructions have been knowis8§& for “Streaming SIMD Extensions.” In its original form,
SSE did not support double-precision floating-point argkio but since the introduction of SSE2 with the

*Copyright© 2010, R. E. Bryant, D. R. O’Hallaron. All rights reserved.



Pentium 4 (2000), SSE provides a viable mechanism for impigimg both single and double-precision
floating-point arithmetic. We will use the term “SSE2+” tond¢e the floating-point support provided by
SSE versions 2 and higher.

All processors capable of executing x86-64 code support2S@Ehigher, and hence x86-64 floating-
point is based on SSE, including conventions for passinggatare arguments and return values [3]. For
IA32, cgcc must be explicitly commanded to generate SSE code using dmtimand-line parameters
‘-mfpmath=sse ’and ‘-msse2 ’(or‘-msse3 ’or higher if the machine supports more recent versions of
SSE.) Even then, the code remains compatible with IA32 autiwes for passing function arguments and
return values.

The media instructions originally focused on allowing rplé operations to be performed in a parallel
mode known asingle instruction, multiple datar SIMD (pronounced SIM-DEE). In this mode the same
operation is performed on a number of different data valogsrallel. The media instructions implement
SIMD operations by having a set of registers that can holdipieldata values ipackedformat. SSE2+
provides either eight (with IA32) or sixteen (with x86-6¥MM registers of 128 bits each, nam@&kmmp
%xmm;land so on, up to either 7 or 15. Each one of these registetsotdia vector ofl’ elements ofV bits
each, such thak’ x N = 128. For integers)N can be 8, 16, 32, or 64 bits, while for floating-point numbers,
N can be 32 or 64. For example, a single SSE instruction canvaaltytte vectors of eight elements each,
while another can multiply two vectors, each containing feingle-precision floating point numbers. The
floating-point formats match the IEEE standard formats iiogle and double-precision values. The major
use of these media instructions are in library routines fapbics and image processing. These routines
can be written in assembly code, or by using special extaadio C supported bgcc, as is covered in
Web Aside OPT:SIMD. There has been considerable effort ablercompilers to extract parallelism from
sequential programs, including tiAeitovectorization Projedn Gcc [4], but so far their capabilities have
proved limited.

With SSE2 came the opportunity to completely change the waatifig-point code is compiled for x86
processors. As described in Web Aside ASM:X87, floating pwias historically implemented in IA32
based on a floating-point architecture dating back to theZ 8@8floating-point coprocessor for the Intel
8086. With this architecture, often referred to as “x87,aflng-point data are held in a shallow stack
of registers, and the floating-point instructions push aop gtack values. This is a difficult target for
optimizing compilers. The architecture also has many gudiie to a nonstandard 80-bit floating-point
format, as described in Web Aside DATA:IA32-FP.

The SSE2+ instructions include a set of instructions to ageonscalar floating-point data, using single
values in the low-order 32 or 64 bits of XMM registers, Thiglat mode provides a set of registers and
instructions that are more typical of the way other processapport floating point. For compilation on
x86-64 and for suitably configured IA32 machineg,c now maps the floating-point data and operations of
a source program into SSE code.

This section describes the implementation of floating pbased on SSE2+. We mostly use x86-64 code in
our examples but also illustrate how code generated for Iég82make use of SSE. Readers may wish to
refer to the Intel documentation for the individual instians [1, 2]. As with integer operations, note that
the ATT format we use in our presentation differs from thelifibrmat used in these documents.



2 Floating-Point Movement and Conver sion Oper ations

Instruction Source | Destination| Description

Movss Msol X X Move single precision

Mmovss X Mss Move single precision

movsd Megal X X Move double precision

movsd X Mgy Move double precision

cvtss2sd M3l X X Convert single to double precision

cvtsd2ss Mgl X X Convert double to single precision

cvtsi2ss Msa/R3o X Convert integer to single precision

cvtsi2sd M3o/R3o X Convert integer to double precision

cvisi2ssq Mgyl Rgy X Convert quadword integer to single precision

cvtsi2sdq Mgyl Rgy X Convert quadword integer to double precision

cvitss2si X/ Mso R3o Convert with truncation single precision to integer
cvitsd2si X/ Mg, R3o Convert with truncation double precision to integer
cvitss2siq X/ Mso Rgy Convert with truncation single precision to quadword irteg
cvitsd2siq X/ Mg, Rgq Convert with truncation double precision to quadword ieteg

X: XMM register (e.g.%xmmp

R3o: 32-bit general-purpose register (e Yheax)
Rg4: 64-bit general-purpose register (e ¥rax)
Mss: 32-bit memory range

Megy: 64-bit memory range

Figure 1:Scalar floating-point movement and conversion operations. These operations transfer values
between memory and registers, possibly converting between data types.

Figure 1 shows a set of instructions for transferring dathfan performing conversions between floating-
point and integer data types. These aresallar instructions, meaning that they operate on individual,
rather than packed, data values. Floating-point data dcedier in memory (indicated in the table as
Mss and Mgy) or in XMM registers (shown in the table as). Integer data are held either in memory
(indicated in the table a&/3, or Mg,4) or in general-purpose registers (shown in the tabl&asand Rg,).
These instructions will work correctly regardless of thigrainent of data, although the code optimization
guidelines recommend that 32-bit memory data satisfy atd-bljgnment, and that 64-bit data satisfy an
8-byte alignment.

The floating-point movement operations can transfer data fiegister to register, from memory to register
and from register to memory. As is true for the integer casingle floating-point instruction cannot move
data from memory to memory. The floating-point conversioarapons have either memory or a register
as source and a register as destination, where the regiséegeneral-purpose registers for integer data and
XMM registers for floating-point data. The instructions¢Buascvttss2si  , for converting floating-point
values to integers use truncation, always rounding valoward zero, as is required by C and most other
programming languages.

As an example of the different floating-point move and cosiger operations, consider the following C
function:



double fevt(int i, float +fp, double +xdp, long *Ip)
{
float f = +fp; double d = xdp; long | = *p;
*Ip = (long) d;
xfp = (float) i;
+dp = (double) I;
return (double) f;
}

and its associated x86-64 assembly code

x86- 64 inplenentation of fcvt

Argunent s:
i Yedi int
fp % si float *
dp % dx doubl e *
Ip % cx long *
1 fevt:
2 movss  (%rsi), %xmmO0
3 movq (%rex), %r8
4 cvttsd2siq (%rdx), %rax
5 movq %rax, (%rcx)
6  Ccvtsi2ss %edi, %xmml
7 movss  %xmm1l, (%rsi)
8 cvtsi2sdq %r8, %xmml
9 movsd %xmm1, (%rdx)

10 cvtss2sd %xmmO, %xmmO

11 ret

Get f = *fp

Get | ==*lp

Get d = »dp and convert to long
Store at Ip

Convert i to float

Store at fp

Convert | to double

Store at dp

Convert f to double

Return f

All of the arguments técvt  are passed through the general-purpose registers, sgcaréeither integers
or pointers. The return value is returned in regi$tetmmpthe designated return register filvat or
double values. In this code, we see a number of the movement andrsomvénstructions of Figure 1.

By comparison, the following is the 1A32 code for body of ftioa fcvt

| A32+SSE i npl enent ati on of fcvt

Argunent s:
i %edp+8 int
fp %ebp+12 float =
dp %ebp+16 doubl e *
Ip %ebp+20 | ong *

1 movl 12(%ebp), %ebx

2 movl 16(%ebp), %esi

3 movl 20(%ebp), %edx

4 movss  (%ebx), %xmml

5 movl (%edx), %ecx

6 cvitsd2si (%oesi), %eax

7 movl %eax, (Yoedx)

8 cvtsi2ss 8(%ebp), %xmmO0
9

movss  %xmm0, (%ebx)

Get fp

Get dp

Get Ip

Get f = *fp

Get | ==*Ip

Get d = »dp and convert to long
Store at dp

Get i and convert to float
Store at *fp



10 cvtsi2sd %ecx, %xmmoO Convert | to double

11 movsd %xmmO0, (%esi) Store at dp

12 cvtss2sd %xmm1, %xmml Convert f to double

13 movsd %xmml, -16(%ebp) Store in nenory

14 fldl -16(%ebp) Read from nenory and push onto x87 stack

The main difference with the 1A32 code is that all argumemes@assed on the stack. The function must
first load the arguments into registers before it can acdesdunction data. Note also the use of the
cvttsd2si instruction (line 6) to convert the double-precision toadgtpelong , whereas the x86-64
code used a&vttsd2siq instruction (line 4). For IA32, botlint andlong are four bytes long. A
final difference is how floating-point values are returnezhfrfunctions, as implemented by instructions
13-14. The x87 floating-point architecture includes a setight floating-point registers organized as a
shallow stack (see ASM:X87). Any floating-point value rekenl from a function should be at the top
of this stack, as implemented by tldl instruction (for double-precision) or ti&ls instruction (for
single-precision.) The only way to transfer data from an XNédister to an x87 register is to first store
it to memory with an SSE instruction (line 13) and then re&ié from memory and push it onto the x87
stack with an x87 instruction (line 14.)

Practice Problem 1.
For the following C code, the expressioradl —val4 all map to the program valués f , d, andl :

double fevt2(int *ip, float *fp, double *dp, long 1)
{
inti = «ip; float f = *fp; double d = *dp;
*ip = (int) vall,;
xfp = (float) val2;
+dp = (double) val3;
return (double) val4;
}

Determine the mapping, based on the following x86-64 codéhfunction:

x86-64 inplenmentation of fcvt2

Argunent s:
ip % di int *
fp % si float =
dp % dx doubl e *
| % cX | ong
1 fevt2:
2 movl (%rdi), %r8d
3 movss  (%rsi), %xmmO
4 cvttsd2si (%rdx), %eax
5 movl %eax, (%rdi)
6  cvtsi2ss %r8d, %xmml
7 movss  %xmml, (%rsi)
8 cvtsi2sdq %rcx, %xmml
9 movsd %xmm1, (%rdx)
10 cvtss2sd %xmmO0, %xmmO

ret

[N
[N



Practice Problem 2:

The following C function converts an argument of tygre _t to a return value of typdst _t , where
these two types are defined usitypedef

dest t cvt(src_t X)

{
dest t y = (dest_t) x;
return y;

}

For execution on x86-64, assume argumeisteither in%oxmm@r in the appropriately named portion of
registerdordi (i.e.,%rdi or%edi), and that one of the conversion instructions is to be usgeétfmrm
the type conversion and to copy the value to the appropyiatahed portion of registérax (integer
result) or%xmm(@floating-point result). Fill in the following table inditiag the instruction, the source
register, and the destination register for the followinghbinations of source and destination type:

Tx Ty Instruction S D
long double | cvtsi2sdq %rdi %xmmO
double int
float double
long float
float long

3 Floating-Point Code in Procedures

With x86-64, the XMM registers are used for passing floapogit arguments to functions and for returning
floating-point values from them. Specifically, the followinonventions are observed:

e Up to eight floating point arguments can be passed in XMM tegg&xmme%xmm./These registers
are used in the order the arguments are listed. Additionatifig-point arguments can be passed on
the stack.

e A function that returns a floating-point value does so insegxmm0o
e All XMM registers are caller saved, The callee may overwdt®g/ of these registers without first

saving it.

When a function contains a combination of pointer, integad floating-point arguments, the pointers and
integers are passed in general-purpose registers, whiftotiting-point values are passed in XMM registers.
This means that the mapping of arguments to registers depandoth their types and their ordering. Here
are several examples:

double fi(int x, double vy, long 2z);



This function would have in %edi, y in %xmmQandz in %rsi .
double f2(double vy, int x, long 2z);

This function would have the same register assignment asifunfl .
double fi(float x, double +xy, long *2);

This function would have in %exmmQy in %rdi , andz in %rsi .

Practice Problem 3:
For each of the following function declarations, deterntimeregister assignments for the arguments:

A. double gil(double a, long b, float c, int d);

B. double g2(int a, double *b, float *C, long d);
C. double g3(double *a, double b, int c, float d);
D double g4(float a, int *b, float c, double d);

4 Floating-Point Arithmetic operations

Single | Double | Effect Description
addss addsd | D <~ D + S Floating-point add
subss subsd | D «~ D — S Floating-point subtract
mulss mulsd | D « D x S Floating-point multiply
divss divsd D+ D/S Floating-point divide
maxss | maxsd | D <« max(D, S) | Floating-point maximum
minss minsd | D < min(D, S) | Floating-point minimum
sqrtss sqrtsd D « /§ Floating-point square rogt
Figure 2:Scalar floating-point arithmetic operations. All have source S and destination D operands.

Figure 2 documents a set of scalar SSE2+ floating-pointuctitms that perform arithmetic operations.
Each has two operands: a sougewhich can be either an XMM register or a memory location, and
destinationD, which must be an XMM register. Each operation has an instnudor single-precision and
an instruction for double precision. The result is storethandestination register.

As an example, consider the following floating-point funaoti

double funct(double a, float x, double b, int i)

{
}

return a *Xx - bl



The x86-64 code is as follows:

x86-64 inplenmentation of funct

Argunent s:
a % D doubl e
X % mmil fl oat
b % doubl e
i %edi int
1 funct:
2 cvtss2sd %xmm1, %xmml Convert x to double
3 mulsd %xmmO0, %xmm1l Mil tiply x by a
4 cvtsi2sd %edi, %xmmO Convert i to double
5 divsd %xmmO0, %xmm2 Conpute b/i
6 movapd %xmml, %xmmO Copy x*a
7 subsd %xmm2, %xmmO Subtract b/ i
8 ret Return

The three floating point argumerdsx, andb are passed in XMM registef&xmme%xmm2while integer
argument is passed in registéxedi. Conversion instructions are required to convert argusverdgnd

i to double (lines 2 and 4.) We will describe thevapd instruction (line 6) in Section 6. Suffice it to
say that, in this case, it copies source regi$t@mmz1o destination registé¥oxmmOThe function value is
returned in registe¥oxmm0

By comparison, refer to the x87 code for this function in #ec6 of Web Aside ASM:X7. Whereas the
x87 code involves operating on the floating-point registacls the SSE code uses registers as individually
addressable storage locations, much as does code operatintgger data.

Practice Problem 4:
For the following C function, the types of the four argumeants defined byypedef

double functl(argl t p, arg2_t q, arg3_t r, argd_t s)
{

}

return p/(g+r) - s;

When compiled for x86-64;cc generates the following code:

x86-64 inplenmentation of functl

1 functl:

2 cvtsi2ss %edi, %xmm3

3 cvisi2ssq %rsi, %xmm2

4 addss %xmm0, %xmm2

5 divss %xmm2, %xmm3

6 cvtss2sd %xmm3, %xmm3
7 movapd %xmm3, %xmmO0

8 subsd %xmm1, %xmmO

9 ret



Determine the possible combinations of types of the founa@nts (there may be more than one.)

Practice Problem 5;
Functionfunct2 has the following prototype

double funct2(double w, int x, float y, long z);
When the function is compiled for x86-6d cc generates the following code:

x86-64 inplenmentation of funct2

1 funct2:

2 cvtsi2ss %edi, %xmm2

3 mulss %xmml, %xmm2

4 cvtss2sd %xmm2, %xmm2
5 cvtsi2sdq %rsi, %xmml

6 divsd %xmm1, %xmmO

7 subsd %xmm0, %xmm2

8 movapd  %xmm2, %xmmO0

9 ret

Write a C version ofunct2

Unlike integer arithmetic operations, the SSE floatingapaperations cannot have immediate values as
operands. Instead, the compiler must allocate and imigéiadtorage for any constant values. The code then
reads the values from memory. This s illustrated by thefeithg Celsius to Fahrenheit conversion function:

double cel2fahr(double temp)
{

}

return 1.8 * temp + 32.0;

The relevant parts of the x86-64 assembly code are as fallows
Code

x86- 64 inpl enmentation of cel 2f ahr
Argunent tenp in %nmD

1 cel2fahr:

2 mulsd .LC2(%rip), Y%oxmmO Miltiply by 1.8

3 addsd .LC3(%rip), %xmmO Add 32.0

4 ret

5 .LC2:

6 .long 3435973837 Low order four bytes of 1.8
7 .long 1073532108 H gh order four bytes of 1.8
8 .LC3:

9 .long 0 Low order four bytes of 32.0
10 .long 1077936128 H gh order four bytes of 32.0

We see that the function reads the value 1.8 from the memaoatit;m labeledLC2 , and the value 32.0
from the memory location labeledd C3 . Looking at the values associated with these labels, wehste t
each is specified by a pair dbng declarations with the values given in decimal. How shoukkéhbe
interpreted as floating-point values? Looking at the detian labeledLC2 , we see that the two values are
3435973837 (Oxcceceeced ) and1073532108 (Ox3ffcccee ). Since the machine uses little-endian byte
ordering, the first value gives the low-order 4 bytes, while second gives the high-order 4 bytes. From
the high-order bytes, we can extract an exponent fielax8ff (1023), from which we subtract a bias of
1023 to get an exponent of 0. Concatenating the fractiondbitee two values, we get a fraction field of
Oxcccecceccecceeced |, which can be shown to be the fractional binary represemtaif 0.8, to which we
add the implied leading one to get 1.8.

Practice Problem 6:
Show how the numbers declared at lalh€l3 encode the number 32.0.

5 Floating-Point Comparison Operations

SSE2+ provides two instructions for comparing floatingapealues:

Instruction Based on| Description
ucomiss S, 51 | S1- Sy | Compare single precision
ucomisd S, 51 | S1- Sy | Compare double precision

These instructions are similar to tbmpl andcmpq instructions (see CS:APP2e Section 3.6), in that they
compare operandS; and S, and set the condition codes to indicate their relative &alues withcmpq,
they follow the ATT-format convention of listing the opedmin reverse order. Argumeft must be in an
XMM register, whileS; can either be in an XMM register or in memory.

The floating-point comparison instructions set three diowlicodes: the zero flagF, the carry flagCF,
and the parity fladPF. We did not document the parity flag in CS:APP2e Chapter Jdmit is not used in
Gcc-generated x86 code. For integer operations, this flag iwiseh the most recent arithmetic or logical
operation yielded a value where the least significant byseslran parity (i.e., an even number of 1's in the
byte). For floating-point comparisons, however, the flagetsrghen either operand idaN. By convention,
any comparison in C is considered to fail when one of the asmimis aNaN, and this flag is used to detect
such a condition. For example, even the comparisor= X yields 0 whernx is aNaN

The condition codes are set as follows:

Ordering | CF ZF PF
Unordered| 1 1 1

< 1 0 0
= 0 1 0
> 0 0 0

The Unorderedcase occurs when either of the operandda$\l. This can be detected from the parity flag.
Commonly, thgp (for “jump on parity”) instruction is used to conditionaljymp when a floating-point

comparison yields an unordered result. Except for this,dasevalues of the carry and zero flags are the
same as those for an unsigned comparigtinis set when the two operands are equal, @ids set when

S1 < Ss. Instructions such g& andjb are used to conditionally jump on various combinations ekth
flags.

As an example of floating-point comparisons, the followindu@ction classifies argumertaccording to
its relation to 0.0, returning an enumerated type as result.

typedef enum {NEG, ZERO, POS, OTHER} range_t;

range_t find_range(float x)

{
int result;
if (x < 0)
result = NEG;
else if (x == 0)
result = ZERO;
else if (x > 0)
result = POS;
else
result = OTHER;
return result;
}

Enumerated types in C are encoded as integers, and so thiel@dssction values are: NEG, 1 (ZERQ,
2 (POS, and 3 OTHER. This final outcome occurs when the valuexat NaN

Gcce generates the following x86-64 code ford _range :

x86- 64 inplenmentation of find_range
Argument x in %mmD (single precision)

1 find_range:

2 mov! $0, %eax Set result = 0

3 ucomiss .LCO(%rip), %xmmO Conpare x:0

4 jp L4 If NaN, goto nonneg

5 jb L7 If <, goto done

6 .L4: nonneg:

7 movl $1, %eax Set result =1

8 ucomiss .LCO(%rip), %xmmO Conpare x: 0

9 jp L8 if NaN, goto posornan
10 e L7 If ==, goto done

11 .L8: posor nan:

12 ucomiss .LCO(%rip), %xmmO Conpare x: 0

13 setbe  %al Set result = NaN? 1: 0
14 movzbl %al, %eax Zero extend

15 addl $2, %eax result += 2 (2 for >0, 3 for NaN)
16 .L7: done:

17 rep ; ret Return result

The code is somewhat arcane—it comparés 0.0 three times, even though the required informatiordcou
be obtained with a single comparison. Let us trace the flowmefunction for the four possible comparison

results.

X < 0.0: Thejb instruction on line 5 will be taken, jumping to the end withedurn value of 0.

x = 0.0: Theje instruction (line 10) will be taken, jumping to the end withreturn value of 1 (set on
line 7.)

X > 0.0: No branches will be taken. Tteetbe (line 13) will yield 0, and this will be incremented by the
addl instruction (line 15) to give a return value of 2.

X = NaN: Bothjp branches (lines 4 and 9) will be taken. Then #etbe instruction (line 13) will
change the return value from 0 to 1, and this value is theremented from 1 to 3 (line 15.)

Compared to the awkward procedure required to extract asidhe floating-point status word with x87
(Web Aside ASM:X87, Section 7), the SSE instructions to d¢towlally compare and test values is very
similar to their counterparts for comparing and testinggets.

Practice Problem 7:
Functionfunct3 has the following prototype

double funct3(int *ap, double b, long c, float *dp);
When the function is compiled for x86-6d cc generates the following code:

x86-64 inplenmentation of funct3

1 funct3:

2 movss  (%rdx), %xmm2

3 cvtsi2sd (%rdi), %xmm1l
4 ucomisd %xmm1, %xmmO

5 jbe L6

6 cvisi2ssq %rsi, %xmmO0
7 mulss %xmm2, %xmmO

8 cvtss2sd %xmm0, %xmmO0
9 ret

10 .L6:

11 cvtsi2ssq %rsi, %xmmO0
12 movaps %xmmz2, %xmml

13 addss %xmm2, %xmml

14 addss %xmm1, %xmmO

cvtss2sd %xmmO, %xmmO
ret

[
o o

Write a C version ofunct3

6 Performing Common Floating-Point Operationsin Uncommon Ways

At times, Gcc makes surprising choices of instructions for performingiowmn tasks. As examples, we've
seen how théeal instruction is often used to perform integer arithmetic (&F¥2e Section 3.5), and the
xorl instruction is used to set registers to 0 (CS:APP2e Problég).3

There are far more instructions for performing floatingrpaperations than we have documented here, and
some of these appear in unexpected places. We document adbwases here.

6.1 Using Instructionsfor Manipulating Packed Data

Instruction | Source| Destination| Description

movaps X X Move aligned, packed single precision

movapd X X Move aligned, packed double precision

cvtps2pd X X Convert packed single to packed double precisian

cvtpd2ps X X Convert packed double to packed single precisian

unpcklps X X Unpack and interleave low packed single precision

unpcklpd X X Unpack and interleave low packed double precision
Figure 3: Some packed format floating-point movement and conversion o perations. These instruc-

tions are often found in scalar code.

Figure 3 shows a number of instructions for manipulating Xividisters containingackedfloating-point
data, where a single XMM register holds either two doubkemion or four single-precision values. We
find these instructions being used in code that operatesan$calar data, low-order value in an XMM
register.

Themovapd andmovaps instructions copy the entire contents of one XMM registeriother. (They can
also copy XMM register contents to and from memory, but wé mat consider these cases here.) We have
already seen instances of thvapd instruction being used to copy from one XMM register to aeath
For these cases, whether the program copies the entirgaregisjust the low-order value affects neither
the program functionality nor the execution speed, and sauhkis instruction rather than the more natural
movsd makes no real difference.

6.2 Converting between Single and Double Precision

Some versions ofcc generate code that uses an idiosyncratic means of conydxéitween single and
double precision values. For example, suppose the lowdode bytes of%oxmmaold a single-precision
value, then the instructioovtss2sd %xmm0, %xmmO would convert it to double precision and store it
in the lower eight bytes dfoxmm0instead, we find the following code generateddnc:

Conversion fromsingle to double precision
1 unpcklps %xmmO0, %xmmO0 Replicate first vector el ement
2 cvtps2pd %xmm0, %xmmO0 Convert two vector elenents to double

The instructionunpcklps instruction is normally used to interleave the values in MMM registers.
That is, if the source register contains wolds, sz, s1, so], and the destination register contains words
[ds,da,dq,dp], then the resulting value of the destination register wdéds, d1, s, dp]. In the code
above, we see that same register being used as source aimatitast and so if the original register held
values|zs, z2, x1, 2¢], then the instruction would update the register to hold eslut, 21, zo, zo]. The
cvtps2pd instruction expands the two low-order single-precisiotues in the source XMM register to
be the two double-precision values in the destination XMMNister. Applying this to the result of the
precedingunpcklps instruction would give valuegixg, dx|, wheredz is the result of converting to
double precision. That is, the net effect of the two insiare is to convert the original single-precision
value in the low-order 4 bytes éfoxmmQ@o double precision and store two copies of it%xmmo0 It is
unclear whyGcc generates this code. There is no benefit or need to have the daplicated within the
XMM register.

Gcc generates similar code for converting from double to sipgéision:

Conversion fromdouble to single precision
1 unpckipd %xmmO0, %xmmO0 Replicate first vector el ement
2 cvtpd2ps %xmm0, %xmmO0 Convert two vector elenents to double

Suppose these instructions start with regi$temm@olding two double-precision valugs;, z;y]. Then the
unpcklpd instruction will set it to[xg, xo]. Thecvtpd2ps will convert these values to single precision,
pack them into the low-order half or the register, and setiigeer half to 0, yielding a result.0, 0.0, g, x|
(recall that floating-point value.0 is represented by a bit pattern of all 0s.) Again, there islearosalue in
computing the conversion from one precision to anothenifaig rather than by using the single instruction
cvtsd2ss %xmmO, %xmmO. The fact that this code is generated only by some versioscafand not
by others seems to indicate that it has no particular benefit.

6.3 Using Bit-Wise Operations

Single | Double | Effect Description
xorps | xorpd | D «+ D~ S | Bit-wise Exclusive-Or
andps |andpd | D + D & S | Bit-wise And

Figure 4:Bit-wise operations on packed data.  These instructions perform Boolean operations on all 128
bits in an XMM register

Figure 4 show that we can perform bitwise operations on XMHyisters, much as we can for the general-
purpose registers. In the code generateasby, we often see these operations being applied to an entire
XMM register, rather than just the low-order 4 or 8 bytes. S&heperations are often simple and convenient
ways to manipulate floating-point values, as is exploredhéfollowing problem.

Practice Problem 8:

Consider the following C function, wheEXPRis a macro defined witkdefine

double simplefun(double x) {
return EXPR(X);
}

Below we show the SSE code generated for different defirstofEXPR where valuex is held in
%xmm0 All of them correspond to some useful operation on floapogit values. Identify what the
operations are. Your answers will require you to understaadit patterns of the constant words being
retrieved from memory.

A. 1 andpd .LC1(%rip), %xmmO
2 .LC1:
3 dong 4294967295
4 dong 2147483647
5 long 0
6 long 0
7 .align 16
B. 1 xorpd  %xmmO0, %xmmO
C. 1 xorpd  .LC2(%rip), %xmmO
2 .LC2:
3 long 0
4 dong  -2147483648
5 long 0
6 long 0

7 Final Observations

We see that the general style of machine code generated éoatopy on floating-point data with SSE is
similar to what we have seen for operating on integer datdh Bse a collection of registers to hold and
operate on values. For x86-64 code, we also use these redmteassing function arguments.

Of course, there are many complexities in dealing with tlileidint data types and the rules for evaluating
expressions containing a mixture of data types, but fund#aliig, SSE code is more straightforward than

the x87 code historically used to implement floating-poipem@tions on x86 machines. In addition, the

SSE code generally runs faster, since there is less needvie data back and forth between registers and
memory.

SSE also has the potential to make computations run fasteetigrming parallel operations on packed
data. Compiler developers are working on automating th@ersion of scalar code to parallel code, but
currently the most reliable way to achieve higher perforoeatihrough parallelism is to use the extensions
to the C language supported By c for manipulating vectors of data. See Web Aside OPT:SIMDe® s
how this can be done.

Solutions to Practice Problems

Problem 1 Solution: [Pg. 5]

This exercise requires that you step through the code, gaareful attention to which conversion and data
movement instructions are used. We can see the values lediigyed and converted as follows:

e The value atp is retrieved, converted to ant (line 4), and then stored g . We can therefore

infer thatvall isd.

e The value atp is retrieved, converted toftoat  (line 6), and then stored & . We can therefore

infer thatval2 isi .

e The value ofl is converted to @ouble (line 8) and stored atp. We can therefore infer thatl3
is|

e The value afp is retrieved, converted todouble (line 10) and left in registe¥oxmm@s the return
value. We can therefore infer the&l4 isf .

Problem 2 Solution: [Pg. 6]
These cases can be handled by selecting the appropriagefrentrthe table in Figure 1.

Tx Ty Instruction S D
long double | cvtsi2sdq %rdi %xmmO
double int cvttsd2si %xmm0O %eax
float double | cvtss2sd %xmmO %xmm0O
long float cvtsi2ssq %rdi %xmm0O
float long cvtss2siq %xmm0O %rax

Problem 3 Solution: [Pg. 7]

The basic rules for mapping arguments to registers arey feimple (although they become much more
complex with more and other types of arguments [3]).

A. double gl(double a, long b, float c, int d);
Registersa in %oxmmpb in %rdi ¢ in %xmm.ld in %esi .
* b, float

B. double g2(int a, double *c, long d);

Registersa in %edi, b in %rsi , ¢ in %rdx, d in %rcx.

C. double g3(double
Registersa in %rdi 0, b in Y%exmmfQc in %esi, d in %oxmml

*a, double b, int c, float d);

D. double g4(float a, int
Registersa in %xmmpb in %rdi , ¢ in %xmm]ld in %oxmm2

*b, float c, double d);

Problem 4 Solution: [Pg. 8]

This problem demonstrates the challenge of matching argtsrte registers when we do not know the
argument data types. One strategy is to work backwards. Wehse the subtraction instruction on line
8 computes the difference of regist@tsxmm@nd %oxmm,where the former contains the result from the
instruction on the preceding line (which must be the restitamputing the expressiop/(g+r) , and
the latter contains argument Tracing back with®oxmmlwe can see that this it must hold the second
floating-point argument, and we can infer tkas of data typedouble .

The result of computing the expressipf{(g+r)  gets converted from single to double precision by the
instruction of line 6, and so we can conclude that among aegusp, g, andr there must be at least one
with data typeloat , and none with data typgouble .

We see that the first two integer arguments are convertedtéotyl@efloat on lines 2 (fromint ) and 3
(from long ). The former is used in the numerator in the division ingtarcof line 5, and hence must be
p, while the second is added to the first floating-point argumen

Fundamentally, however, there is no way to determine tlaivel orderings of the first floating-point argu-
ment and the second integer argument, since the additiamatope of line 4 is commutative. One must be
argumenip and the other must be argument q, but there is no way to distihdgetween the two.

In fact, Gcc generates the exact came code when the argumefusafi. are defined according to either
of the following two prototypes:

double functla(int p, float g, long r, double s);
double functlb(int p, long q, float r, double s);

Problem 5 Solution: [Pg. 9]

This problem can readily be solved by stepping through tiserably code and determining what is com-
puted on each step, as shown with the annotations below:

1 funct2:
x86-64 inpl ementation of funct2
Argunent s:
w %D  doubl e
X Yedi int
y % mil fl oat
z % si | ong
2 cvtsi2ss %edi, %xmm2 Convert x to float
3 mulss %xmm1, %xmm2 Ml tiply by y
4 cvtss2sd %xmm2, %xmm?2 Convert x*y to double
5 cvtsi2sdq %rsi, %xmml Convert z to double
6 divsd %xmm1, %xmmO Conpute w z
7 subsd %xmmO0, %xmm2 Conpute x*y-w z
8 movapd %xmm2, %xmmO0
9 ret

We can conclude from this analysis that the function congpysex - w/z .

Problem 6 Solution: [Pg. 10]

This problem involves the same reasoning as was requiregedrat numbers declared at labeC2
encode 1.8, but with a simpler example.

We see that the two values are 0 and 10779360280400000 ). From the high-order bytes, we can
extract an exponent field ®@x404 (1028), from which we subtract a bias of 1023 to get an expiooieb.
Concatenating the fraction bits of the two values, we gegetion field of 0, but with the implied leading
value giving value 1.0. The constant is therefore x 2° = 32.0.

Problem 7 Solution: [Pg. 12]
Again, we annotate the code, including dealing with the @@l branch:

x86-64 inpl enmentation of funct3

Argunent s:
ap % di int *
% D doubl e
% si | ong
dp % dx doubl e *
1 funct3:
2 movss  (%rdx), %xmm2 Get d = »dp
3 cvtsi2sd (%rdi), %xmm1l Get a = xap and convert to double
4 ucomisd %xmm1, %xmmO0 Conpare b: a
5 jbe .L6 If b <= a, goto |esseq
6 cvtsi2ssq %rsi, %xmmO0 Convert ¢ to float
7 mulss %xmm2, %xmmO Ml tiply by d
8 cvtss2sd %xmmO, %xmmO Convert c+d to double
9 ret Return
10 .L6: | esseq:
11 cvtsi2ssq %rsi, %xmmO0 Convert ¢ to float
12 movaps %xmmz2, %xmml Copy d
13 addss %xmm2, %xmm1l Conpute d+d = 2xd
14 addss %xmm1l, %xmmO Conpute ¢ + 2*d
15 cvtss2sd %xmmO0, %xmmO Convert to double
16 ret Return

From this, we can write the following code farnct3

double funct3(int

int a = =*ap;
float d =  =dp;
if (@ < b)

return ¢ *d;
else
return c+2 =d;

Problem 8 Solution: [Pg. 14]

+xap, double b, long c, float

*dp) {

A. We see here that the 16 bytes starting at addieS4 form a mask, where the low-order 8 bytes
contain all 1's, except for most significant bit, which is #ign bit of a double-precision value. When

we compute theND of this mask with%xmmpit will clear the sign bit ofx, yielding the absolute
value. In fact, we generated this code by defifiXPR(X) to befabs(x) , wherefabs is defined
in <math.h> .

B. We see that thetorpd instruction sets the entire register to 0, and so this is a twwagenerate
floating-point constani.o.

C. We see that the 16 bytes starting at addrie€2 form a mask with a single one bit, at the position of
the sign bit for the low-order value in the XMM register. Wheae compute th&xCLUSIVE-OR of
this mask with%oxmmpwe change the sign of, computing the expression .

NaN, floating-point not-a-number, 10 floating-point architecturel,
CS:APP2e, 1
%xmmQReturn floating-point value register, 4  Intel format assembly code, 2

%xmm@preturn floating-point value, 6
8086, Intel 16-bit microprocessor, 2

8087, floating-point coprocessor, 2 MMX, Intel media extension, 1

movapd [x86-64] Move aligned, packed double pre-
cision, 13

movaps [x86-64] Move aligned, packed single pre-
cision, 13

movsd [x86-64] Move double precision, 3

movss [x86-64] Move single precision, 3

jp [x86-64] Jump when parity flag set, 10

andpd [x86-64] And packed double precision, 14
andps [x86-64] And packed single precision, 14
architecture

floating-point,1
ATT format assembly code, 2
ATT format, argument listing, 10

cvtpd2ps  [x86-64] Convert packed double to pacIPee}:fked data format, 2
' isi packed format data, 13
single precision, 13

cvtps2pd [x86-64] Convert packed single to packagdalar format date, 13

double precision, 13, 14 SIMD, Single-Instruction Multiple Data execution, 2
cvtsd2ss  [x86-64] Convert single to double preSSE, Streaming SIMD Extensiorts,

cision, 3 SSE2+, SSE versions 2 and higher, 2
cvtsi2sd  [x86-64] Convert integer to double pre-
cision, 3 ucomisd [x86-64] compare double precision, 10
cvtsi2sdq  [x86-64] Convert quadword integer taucomiss [x86-64] compare single precision, 10
double precision, 3 Unordered, floating-point comparison outcome, 10
cvtsi2ss  [x86-64] Convert integer to single preunpcklpd [x86-64] Unpack and interleave low packed
cision, 3 double precision, 13, 14
cvtsi2ssq  [x86-64] Convert quadword integer taunpcklps [x86-64] Unpack and interleave low packed
single precision, 3 single precision, 13, 14
cviss2sd  [x86-64] Convert single to double pre- ) o ]
cision, 3 x87, floating-point instructions for 8087, 2

cvitsd2si  [x86-64] Convert double precision to*MM, SSE2+ registers, 2 _
xorpd [x86-64] Exclusive-or packed double preci-

integer, 3 :
cvttsd2siq  [x86-64] Convert double precision to sion, 14 _ _ _
quadword integer, 3 xorps [x86-64] Exclusive-or packed single preci-

sion, 14

cvttss2si [x86-64] Convert single precision to
integer, 3

cvttss2siq [x86-64] Convert single precision to
guadword integer, 3

fldd [IA32/x86-64] x87 FP load double precision,
5
flds [IA32/x86-64] x87 FP load single precision, 5

